*============The_Clipping_PM_Spectrum================================

Two real sidebands at 50% of carrier will AM at +/-100%.
Two equal size imaginary sidebands will PM at about +/- 1 radian.

One would not expect clipping to effect PM.
But the extra side bands that get created are worth a look.

```
PM_Clipping
* dsauersanjose@aol.com  1/18/20
* www.idea2ic.com                  replace(OPT-SPACE)=>SPACE
*
*    |OUT|    .5sidebands should product +/- 1.0radians PM
*     |_|     .3sidebands should product +/- 0.6radians PM
*     _|_     +/- (.6/6.28318)*100ms =   +/- 9.5msec
*    /   \
*   //_\  \
*   \    \//
*    \___/
*     _|_
*     ///
*
*=========The_SideBand_Tests=========================
VT          VT      0       PWL(    0       0       2       2 )    DC      0
VP          VP      0       DC      3.141592653589793
V2P         V2P     0       DC      6.283185307179586
BREF        VREF    0       V =     cos( v(V2P)*10*v(VT))
BPM1        VPM1    0       V =     cos( v(V2P)*10*v(VT)) -.3*cos( v(V2P)*9*v(VT)) +.3*cos( v(V2P)*11*v(VT))
BPM2        VPM2    0       V =     cos( v(V2P)*10*v(VT)) +.3*sin( v(V2P)*9*v(VT)) +.3*sin( v(V2P)*11*v(VT))
*=====================================================================
```

The carrier is modulated at +/-0.6 radians => 68_deg_ppk




```
*=========Clipping_Circuit=========================
BREFC       VREFC   0       V =     2*u( v(VREF))-1
BPM1C       VPM1C   0       V =     2*u( v(VPM1))-1
BPM2C       VPM2C   0       V =     2*u( v(VPM2))-1
*=========Run_Sim==========================
.control
set         pensize = 2
*TRAN       TSTEP   TSTOP   TSTART TMAX    ?UIC?
tran        1m      2       0       1m
*============PlotResults=============================================
plot vpm1     vpm2
plot vpm1c    vpm2c
*=====================================================================
```

All PM happens at the zero crossings.




```
*===========Find_Spectrum_VPM1=========================
linearize
```

```
set          specwindow =    "none"
*SPEC        FSTART  FSTOP    FSTEP   VECTOR
spec         .5      500      .5      v(VPM1)
plot         mag(v(VPM1))   loglog   ylimit 1m 1
*=============================================================
```

Not a bad idea to look at the pre-clipping spectrum.




```
*=========ReRun_Simulation==========================
*TRAN      TSTEP   TSTOP   TSTART TMAX    ?UIC?
tran       1m      2       0      1m
*=============================================================
```

Need to rerun to get VPM1C.

```
*===========Find_Spectrum_VAM1C=========================
linearize
set          specwindow =    "none"
*SPEC        FSTART  FSTOP    FSTEP   VECTOR
spec         .5      500      .5      v(VPM1C)
plot         mag(v(VPM1C))   loglog   ylimit 1m 1
*=============================================================
```



Now what is each side band doing?
Print out the spectrum to see the results.

```
*============Printout_Spectrum_VPM1C=====================
echo         VPM1C spectrum
foreach ii   6 7 8 9 10 11 12 13 14 26 27 28 29 30 31 32 33 34
let i =      $ii
```

```
let fr=      frequency[2*i-1]
let vfftr =  real(v(VPM1C)[2*i-1])
let vffti =  imag(v(VPM1C)[2*i-1])
echo freq=   $&fr   real=  $&vfftr  imag=  $&vffti
end
.endc
.end
*=============================================================
```

The following is the MacSpice Printout.

```
VPM1C      spectrum
freq = 6   real = 0.001079     imag = 0.00786568
freq = 7   real = -0.012364    imag = 4.00106E-05
freq = 8   real = 0.0445102    imag = 0.0015903
freq = 9   real = -0.339162    imag = -0.00865536
freq = 10  real = 1.17715      imag = 0.00930742
freq = 11  real = 0.338821     imag = -0.00319551
freq = 12  real = 0.045779     imag = -0.00100901
freq = 13  real = 0.0123131    imag = -0.00276141
freq = 14  real = 0.00243758   imag = 0.00214823
freq = 26  real = -0.00920736  imag = -0.00950437
freq = 27  real = 0.0383328    imag = -0.00245988
freq = 28  real = -0.109151    imag = -0.00782071
freq = 29  real = 0.245864     imag = 0.00996243
freq = 30  real = -0.17677     imag = -0.00425125
freq = 31  real = -0.244878    imag = -0.00110665
freq = 32  real = -0.110234    imag = -2.30459E-05
freq = 33  real = -0.0378836   imag = 0.00243613
freq = 34  real = -0.0114339   imag = 0.00385237
```

As expected, the PM sidebands are about the same.
The PM sidebands are the asymmetrical real ones.

All the square wave odd harmonics also see PM.

And there is a little AM as well, this is perhaps to maintain the square shape.

For now, color coding PM side bands versus AM sidebands helps make things stand out.

==========Full_Netlist_For_Copy_Paste==========

```
PM_Clipping
* dsauersanjose@aol.com  2.10.10_3.50PM
* www.idea2ic.com                replace(OPT-SPACE)=>SPACE
*
*    ____
*   |OUT |    .5sidebands should product +/- 1.0radians PM
*   |____|    .3sidebands should product +/- 0.6radians PM
*     _|_     +/- (.6/6.28318)*100ms =   +/- 9.5msec
*    /_ _\
*   //   \\
*   \\   //
*    \___/
*     |
*    /|\
*
*V_PULSE# NODE_P NODE_N PULSE( VINIT  VPULSE TDELAY TRISE  TFALL  PWIDTH PERIOD )
VIN       VIN     0      PULSE( 0      1      .25    1u     1u     .5     1      )
*=========The_SideBand_Tests=========================
VT        VT      0      PWL(   0      0      2       2 )   DC     0
VP        VP      0      DC     3.141592653589793
V2P       V2P     0      DC     6.283185307179586
BREF      VREF    0      V =    cos( v(V2P)*10*v(VT))
BPM1      VPM1    0      V =    cos( v(V2P)*10*v(VT)) -.3*cos( v(V2P)*9*v(VT)) +.3*cos( v(V2P)*11*v(VT))
BPM2      VPM2    0      V =    cos( v(V2P)*10*v(VT)) +.3*sin( v(V2P)*9*v(VT)) +.3*sin( v(V2P)*11*v(VT))
*=========Clipping_Circuit=========================
BREFC     VREFC   0      V =    2*u( v(VREF))-1
BPM1C     VPM1C   0      V =    2*u( v(VPM1))-1
BPM2C     VPM2C   0      V =    2*u( v(VPM2))-1
*=========Run_Sim=========================
.control
set       pensize = 2
```

```
*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
tran      1m     2      0       1m
*============PlotResults=============================================
plot vpm1    vpm2
plot vpm1c   vpm2c
*=========Find_Spectrum_VPM1=========================
linearize
set       specwindow =    "none"
*SPEC     FSTART  FSTOP    FSTEP   VECTOR
spec      .5      500     .5       v(VPM1)
plot      mag(v(VPM1))   loglog   ylimit 1m 1
*=========ReRun_Simulation=======================
tran      1m     2      0       1m
*===========Find_Spectrum_VPM1C=======================
linearize
set         specwindow =    "none"
*SPEC       FSTART  FSTOP    FSTEP   VECTOR
spec        .5      500     .5       v(VPM1C)
plot        mag(v(VPM1C))   loglog  ylimit 1m 1

*============Printout_Spectrum_VPM1C=======================
echo         VPM1C spectrum
foreach ii   6 7 8 9 10 11 12 13 14   26  27 28 29 30 31 32 33 34
let i =      $ii
let fr=      frequency[2*i-1]
let vfftr =  real(v(VPM1C)[2*i-1])
let vffti =  imag(v(VPM1C)[2*i-1])
echo freq=   $&fr   real=  $&vfftr  imag=  $&vffti
end
.endc
.end
```

2.12.10_2.51PM
dsauersanjose@aol.com
Don Sauer
http://www.idea2ic.com/